Handout for

Intervention Outcome Evaluation: Using Structural Equation Models to Understand Change in Youth Interventions

Useful Resources

Model Building

Jaccard, J. (2011). Theory construction, model building, and model selection. In T. Little (Ed.) *Handbook of quantitative methods*. New York: Oxford.

Jaccard, J., & Jacoby, J. (2010). *Theory construction and model-building skills: A practical guide for social scientists*. New York: Guilford.

Confirmatory Factor Analysis

Brown, T. A. (2015). *Confirmatory factor analysis for applied research*. New York: Guilford.

Structural Equation Modeling

Kline, R. B. (2015). *Principles and practice of structural equation modeling*. New York: Guilford.

Little, P. T. D. (2013). *Longitudinal structural equation modeling*. New York: Guilford. [***irreplaceable resource, new edition coming***]

Bayesian Structural Equation Modeling

Zyphur, M. J., & Oswald, F. L. (2015). Bayesian estimation and inference: A User's Guide. *Journal of Management*, 41, 390-420. doi: 0149206313501200.

Mplus

Geiser, C. (2012). *Data analysis with Mplus*. New York: Guilford. [***best place to start with Mplus***]

Heck, R. H., & Thomas, S. L. (2015). *An Introduction to Multilevel Modeling Techniques: MLM and SEM Approaches Using Mplus*. New York: Routledge.

Muthén, L. K., & Muthén, B. O. (2012). Mplus User's Guide (7th edn).

Wang, J., & Wang, X. (2012). *Structural equation modeling: Applications using Mplus*. Hoboken, NJ: John Wiley & Sons.

Basic Mplus Syntax

TITLE:
Optional. Best to provide detailed description of analysis for future reference. No line or character limit.

DATA: Tells Mplus where to get the data file.
file =

VARIABLE: Defines variable names and variable types.
NAMES ARE: x y m z; Defines the variable names and variable types.
USEVARIABLES ARE: x y m; Specifies subset of variables to use.
MISSING ARE ALL (999); Missing values are defined by 999 in dataset.
CATEGORICAL ARE y; Specifies which variables are ordered categorical;
CLUSTER IS school; Specifies cluster variable (e.g., school building).

ANALYSIS: Defines estimator and type of analysis.
ESTIMATOR = mlr; Specifies type of estimator (e.g., robust ml).
TYPE = twolevel; Specifies type of analysis (e.g., multilevel).

MODEL: Defines the model to be analyzed.
BY "Measured by" defines latent variables (e.g., **f1 by x1 x2 x3;**).
ON "Regressed on" defines regression equations (e.g., **y on x;**).
WITH "Correlated with" defines a correlation (e.g., **x with y;**).

OUTPUT: Requests specific output.
SAMPSTAT Provides basic descriptive statistics.
STANDARDIZED Provides the standardized solution.
RESIDUAL Provides predicted covariances and undstandardized residuals.
CINTERVAL Provides confidence intervals.
MODINDICES(ALL) Provides all modification indices.

MODEL 1



MODEL 1 in Mplus

```
TITLE:  Model 1 Single Indicator Two Waves
DATA:  file  = file location.dat

VARIABLE:
NAMES ARE cluster interv med1 med2 med3 out1 out2 out3;
USEVARIABLES ARE interv med1 med2 out1 out2;
MISSING = all(999);

ANALYSIS:
ESTIMATOR = mlr;

MODEL:
med2 on med1 interv;
out2 on out1 med1 med2 interv;

MODEL INDIRECT:
out2 IND interv;

OUTPUT: sampstat standardized cinterval;
```

```
TITLE:  Model 1 Single Indicator Two Waves, Bootstrapped Confidence Intervals
DATA:  file  = file location.dat

VARIABLE:
NAMES ARE cluster interv med1 med2 med3 out1 out2 out3;
USEVARIABLES ARE interv med1 med2 out1 out2;
MISSING = all(999);

ANALYSIS:
BOOTSTRAP = 1000;

MODEL:
med2 on med1 interv;
out2 on out1 med1 med2 interv;

MODEL INDIRECT:
out2 IND interv;

OUTPUT: sampstat standardized cinterval(BCBOOTSTRAP);
```

```
TITLE:  Model 1 Single Indicator Two Waves, Bayesian Estimator,
Noninformative Priors
DATA:  file  = file location.dat

VARIABLE:
NAMES ARE cluster interv med1 med2 med3 out1 out2 out3;
USEVARIABLES ARE interv med1 med2 out1 out2;
MISSING = all(999);

ANALYSIS:
ESTIMATOR = bayes;
PROCESSORS = 2;
FBITERATIONS = 20000;

MODEL:
med2 on med1;
med2 on interv (a);
out2 on out1 med1 interv;
out2 on med2 (b)

MODEL CONSTRAINT:
NEW (ab);
ab = a*b;

OUTPUT: sampstat standardized cinterval;
```

MODEL 2


e1
e2
e3
M1a
M1b
M1c
Mediator T1
e7
e8
e9
M2a
M2b
M2c
d1
Mediator T2
A
Intervention vs. Control
B
C
Outcome T2
d2
O2a
O2b
O2c
e10
e11
e12
Outcome T1
O1a
O1b
O1c
e4
e5
e6

MODEL 2 in Mplus

```
TITLE:  Model 2 Multiple Indicators Two Waves
DATA:  file  = file location.dat

VARIABLE:
NAMES ARE cluster interv med1a med1b med1c med2a med2b med2c
med3a med3b med3c out1a out1b out1c out2a out2b out2c out3a
out3b out3c;
USEVARIABLES ARE interv med1a med1b med1c med2a med2b med2c
out1a out1b out1c out2a out2b out2c;
MISSING = all(999);

ANALYSIS:
ESTIMATOR = mlr;

MODEL:
!measurement model (equal form, equal factor loadings, equal intercepts)
med1 by med1a* (a) !specifies equal factor loading across time
med1b (b)
med1c (c);
med1@1;
med2 by med2a* (a)
med2b (b)
med2c (c);
med2@1;
out1 by out1a* (d)
out1b (e)
out1c (f);
out1@1;
out2 by out2a* (d)
out2b (e)
out2c (f);
out2@1;

[med1a@0]; [med2a@0]; [out1a@0]; [out2a@0]; !fix 1st indicator intercept at 0
[med1*]; [med2*]; [out1*]; [out2*]; !estimate latent factor means
[med1b med2b] (1); [med1c med2c] (2); !fix indicator intercepts to be equal
[out1b out2b] (3); [out1c out2c] (4); !fix indicator intercepts to be equal

!structural model
med2 on med1 interv;
out2 on out1 med1 med2 interv;

MODEL INDIRECT:
Out2 IND interv;

OUTPUT: sampstat standardized cinterval;
```

MODEL 3


Mediator
T1
e1
e3
Mediator
T2
Mediator
T3
A
B
Intervention
vs. Control
C
Outcome
T2
Outcome
T3
e2
e4
Outcome
T1

MODEL 3 in Mplus

```
TITLE:  Model 3 Single Indicator Three Waves
DATA:  file  = file location.dat

VARIABLE:
NAMES ARE cluster interv med1 med2 med3 out1 out2 out3;
USEVARIABLES ARE interv med1 med2 med3 out1 out2 out3;
MISSING = all(999);

ANALYSIS:
ESTIMATOR = mlr;

MODEL:
med3 on med1 med2 interv;
med2 on med1 interv;
out3 on out1 out2 med1 med2 med3 interv;
out2 on out1 med1 med2 interv;

MODEL INDIRECT:
Out3 IND interv;

OUTPUT: sampstat standardized cinterval;
```

MODEL 4


e1
e2
e3
M1a
M1b
M1c
Mediator
T1
e7
e8
e9
M2a
M2b
M3c
e13
e14
e15
M3a
M3b
M3c
d1
d3
Mediator
T2
Mediator
T3
A
Intervention
vs. Control
B
C
Outcome
T2
Outcome
T3
d2
d4
O2a
O2b
O2c
Outcome
T1
O3a
O3b
O3c
e10
e11
e12
e16
e17
e18
O1a
O1b
O1c
e4
e5
e6

MODEL 4 in Mplus

```
TITLE:  Multiple Indicators Three Waves
DATA:  file  = file location.dat
VARIABLE:
NAMES ARE cluster interv med1a med1b med1c med2a med2b med2c med3a med3b
med3c out1a out1b out1c out2a out2b out2c out3a out3b out3c;
USEVARIABLES ARE interv med1a med1b med1c med2a med2b med2c med3a med3b med3c
out1a out1b out1c out2a out2b out2c out3a out3b out3c;
MISSING = all(999);
ANALYSIS:
ESTIMATOR = mlr;

MODEL:
!measurement model (equal form, equal factor loadings, equal intercepts)
med1 by med1a* (a) !specifies equal factor loading across time
med1b (b)
med1c (c);
med1@1;
med2 by med2a* (a)
med2b (b)
med2c (c);
med2@1;
med3 by med2a* (a)
med2b (b)
med2c (c);
med3@1;
out1 by out1a* (d)
out1b (e)
out1c (f);
out1@1;
out2 by out2a* (d)
out2b (e)
out2c (f);
out2@1;
out3 by out2a* (d)
out2b (e)
out2c (f);
out3@1;

!fix 1st indicator intercept at 0, estimate factor means,
!equal intercepts across time
[med1a@0]; [med2a@0]; [med3a@0]; [out1a@0]; [out2a@0]; [out3a@0];
[med1*]; [med2*]; [med3*]; [out1*]; [out2*]; [out3*];
[med1b med2b med3b] (1); [med1c med2c med3c] (2);
[out1b out2b out3b] (3); [out1c out2c out3c] (4);

!structural model
med3 on med1 med2 interv;
med2 on med1 interv;
out3 on out1 out2 med1 med2 med3 interv;
out2 on out1 med1 med2 interv;

MODEL INDIRECT:
Out3 IND interv;

OUTPUT: sampstat standardized cinterval;
```

MODEL 5



Organization



MODEL 5 in Mplus

```
TITLE:  Model 5 Multilevel Single Indicator Two Waves
DATA:  file  = file location.dat

VARIABLE:
NAMES ARE cluster interv med1 med2 med3 out1 out2 out3;
USEVARIABLES ARE interv med1 med2 out1 out2;
MISSING = all(999);
CLUSTER = cluster; !specifies the cluster variable
BETWEEN = interv; !specifies variables that only have level 2 variance
WITHIN = med1 out1;

ANALYSIS:
TYPE = TWOLEVEL; !specifies a multilevel analysis with two levels
ESTIMATOR = mlr;

MODEL:
%within% !level 1 (individual level) model
med2 on med1;
out2 on out1 med1 med2;

%between% !level 2 (organization level) model
med2 on interv (a);
out2 on interv;
out2 on med2 (b);

MODEL CONSTRAINT:
NEW (indab);
indab = a*b;

OUTPUT: sampstat standardized cinterval;
```

```
TITLE:  Model 5 Single Level Alternative to Multilevel Model
DATA:  file  = file location.dat

VARIABLE:
NAMES ARE cluster interv med1 med2 med3 out1 out2 out3;
USEVARIABLES ARE interv med1 med2 out1 out2;
MISSING = all(999);
CLUSTER = cluster; !specifies the cluster variable

ANALYSIS:
TYPE = COMPLEX; !invokes Huber-White sandwich estimator to correct SEs
ESTIMATOR = mlr;

MODEL:
med2 on med1 interv;
out2 on out1 med1 med2 interv;

MODEL INDIRECT:
out2 IND interv;

OUTPUT: sampstat standardized cinterval;
```

MODEL 6


Mediator T1
Mediator T2
Mediator T3
1
1
1
d1
d2
Mediator Change 1
Mediator Change 2
A
Intervention vs. Control
B
C
Outcome Change 1
Outcome Change 2
d3
d4
1
1
1
Outcome T2
Outcome T3
Outcome T1

MODEL 6 in Mplus

```
TITLE:  Model 6 Latent Change Model Three Waves
DATA:  file  = file location.dat

VARIABLE:
NAMES ARE cluster interv med1 med2 med3 out1 out2 out3;
USEVARIABLES ARE interv med1 med2 med3 out1 out2 out3;
MISSING = all(999);

ANALYSIS:
ESTIMATOR = mlr;

MODEL:
!specify latent change for mediator
medch1 by med2@1 med3@1;
medch2 by med2@0 med3@1;
medch1 with medch2*;
med2@0; med3@0; !Fix error variances at zero
[med2@0]; [med3@0]; !Fix means/intercepts at zero
medch1*; medch2*; !Estimate variances
[medch1*]; [medch2*]; !Estimate means

!specify latent change for outcome
outch1 by out2@1 out3@1;
outch2 by out2@0 out3@1;
outch1 with outch2*;
out2@0; out3@0; !Fix error variances at zero
[out2@0]; [out3@0]; !Fix means/intercepts at zero
outch1*; outch2*; !Estimate variances
[outch1*]; [outch2*]; !Estimate means

med2 med3 on med1; !covary out time 1
out2 out3 on out1; !covary out time 1
medch1 on interv (a);
medch2 on interv;
outch1 on interv medch1;
outch2 on interv medch1 medch2;
outch2 on medch1 (b);

MODEL CONSTRAINT:
NEW (ab);
ab = a*b;

OUTPUT: sampstat standardized cinterval;
```